

جذور المعادلات الغير خطية
مثال: أوجد جذور المعادلة التربيعية الآتية

$$x^2 - 24.13x + 1.40 = 0$$

طريقة نيوتن:

لحل المعادلة أعلاه بطريقة نيوتن نبدأ بنقطة في جوار الجذر ولتكن $x_0=0$ حيث أن

$$f(x) = x^2 - 24.13x + 1.40$$

$$f'(x) = 2x - 24.13$$

$$x_1 = x_0 - \frac{f(x_0)}{f'(x_0)}$$

ندخل المعادلة ومشتقتها من أجل تطبيق صيغة نيوتن

	A	B	C
1	x	f	f'
2	0	=A2^2-24.13*A2+1.4	=2*A2-24.13
3	=A2-B2/C2		
4			

بعدها نسحب الخلية B2 إلى B3 بأداة التعبئة التلقائية Fill handle tool

نسحب الخلية C2 إلى C3 بأداة التعبئة

نكرر عملية ولكن للخلايا A4, B4, and C4 معا حتى نحصل على الدقة المطلوبة

	A5	B5	C5
1	x	f	f'
2	0	1.4	-24.13
3	0.058019	0.003366	-24.014
4	0.058159	1.96E-08	-24.0137
5	0.058159	0	-24.0137

لو بدئنا بالحل بطريقة نيوتن ولكن في جوار الجذر الثاني للمعادلة السابقة وهي النقطة $x_0=20$ وقمنا بنفس الخطوات السابقة لوجدنا الحل كما في الجدول

	A6	B6	C6
1	x	f	f'
2	20	-81.2	15.87
3	25.11657	26.17931	26.10314
4	24.11365	1.005844	24.09731
5	24.07191	0.001742	24.01383
6	24.07184	5.26E-09	24.01368

طريقة القاطع:

لحل المعادلة أعلاه بطريقة القاطع نبدأ بالنقاط 0، 1 حيث أن $x_1=1, x_0=0$

$$f(x) = x^2 - 24.13x + 1.40$$

$$x_2 = x_1 - \frac{(x_1 - x_0)f(x_1)}{f(x_1) - f(x_0)}$$

ندخل المعادلة والقيم الأولية

	A	B	C	D	E
1	x0	x1	f(x0)	f(x1)	x2
2	0	1	=A2^2-24.13*A2+1.4	=B2^2-24.13*B2+1.4	=B2-((B2-A2)*D2/(D2-C2))
3					

في الخلية A3 نكتب =B2

في الخلية B3 نكتب =E2

بأداة التعبئة نسحب بمقدار خلية للأسفل بعد تظليل الخلايا C3, D3, E3
بأداة التعبئة نظل النطاق A3:E3 ونسحب حتى نحصل على الدقة المطلوبة

E5		fx		=B5-((B5-A5)*D5/(D5-C5))	
	A	B	C	D	E
1	x0	x1	f(x0)	f(x1)	x2
2	0	1	1.4	-21.73	0.060527
3	1	0.0605	-21.73	-0.0568639	0.058063
4	0.0605	0.0581	-0.057	0.00232178	0.058159
5	0.0581	0.0582	0.0023	-2.29E-07	0.058159

طريقة التنصيف:

لحل المعادلة أعلاه بطريقة التنصيف على الفترة [0,1] للدالة

$$f(x) = x^2 - 24.13x + 1.40$$

ندخل البيانات والمعادلات في ورقة العمل كما يلي

	A	B	C	D	E	F	G
1	a	c	b	fa	fb	fc	check
2	0	=(A2+C2)/2	1	=A2^2-24.13*A2+1.4	=B2^2-24.13*B2+1.4	=C2^2-24.13*C2+1.4	=D2*E2
3	=IF(G2<0,A2,B2)	=(A3+C3)/2	=IF(G2<0,B2,C2)				
4							

نظل الخلايا في النطاق d2:g2 ونسحب بأداة التعبئة للأسفل بمقدار صف

نظل النطاق الجديد وهو A3:G3 ونسحب حتى نحصل على الدقة المطلوبة للجذر وهي c في ورقة العمل

B13		fx		=(A13+C13)/2				
	A	B	C	D	E	F	G	H
1	a	c	b	fa	fb	fc	check	
2	0	0.5	1	1.4	-10.415	-21.73	-14.581	
3	0	0.25	0.5	1.4	-4.57	-10.415	-6.398	
4	0	0.125	0.25	1.4	-1.60063	-4.57	-2.24088	
5	0	0.0625	0.125	1.4	-0.10422	-1.60063	-0.14591	
6	0	0.03125	0.0625	1.4	0.646914	-0.10422	0.90568	
7	0.03125	0.046875	0.0625	0.646914	0.271104	-0.10422	0.175381	
8	0.046875	0.054688	0.0625	0.271104	0.083381	-0.10422	0.022605	
9	0.054688	0.058594	0.0625	0.083381	-0.01043	-0.10422	-0.00087	
10	0.054688	0.056641	0.058594	0.083381	0.03647	-0.01043	0.003041	
11	0.056641	0.057617	0.058594	0.03647	0.013017	-0.01043	0.000475	
12	0.057617	0.058105	0.058594	0.013017	0.001291	-0.01043	1.68E-05	
13	0.058105	0.05835	0.058594	0.001291	-0.00457	-0.01043	-5.9E-06	
14								

قيمة الجذر
C=0.05835

طريقة النقطة الثابتة:

لحل المعادلة أعلاه بطريقة النقطة الثابتة نعيد صياغة المسألة ونبدأ بالنقطة $x=0$

$$f(x) = x^2 - 24.13x + 1.40$$

$$x = \frac{1.40}{x - 24.13}$$

ندخل البيانات والمعادلات

A5		fx	
	A	B	C
1	x	gx	
2	0	=-1.4/(A2-24.13)	
3	=B2		
4			

بعد إدخال صيغة النقطة الثابتة حسب الورقة أعلاه نظل الخلية B2 ونسحبها إلى B3 بأداة التعبئة fill handle ونظل بعدها النطاق A3:B3 ونسحب حتى نحصل على الدقة المطلوبة

B4		fx		=-1.4/(A4-24.13)			
	A	B	C	D	E	F	G
1	x	gx					نت
2	0	0.058019					
3	0.058019	0.058159					
4	0.058159	0.058159					

الطرق المباشرة لحل منظومة خطية من المعادلات الجبرية

طريقة النظير الضربي

لو أردنا حل المنظومة الآتية:-

$$x_1 - x_2 + 2x_3 - x_4 = -8$$

$$2x_1 - 2x_2 + 3x_3 - 3x_4 = -20$$

$$x_1 + x_2 + 2x_3 = -2$$

$$x_1 - x_2 + 4x_3 + 3x_4 = 4$$

ندخل بيانات المنظومة في ورقة العمل كما يلي:-

	A	B	C	D	E	F
1		x1	x2	x3	x4	b
2		1	-1	2	-1	-8
3		2	-2	3	-3	-20
4		1	1	1	0	-2
5		1	-1	4	3	4
6						

نستعمل الدالة minverse لحساب نظير المصفوفة كما يلي

نحدد منطقة فارغة بحجم 4×4 من أجل أن نحسب المصفوفة النظيرة ثم نستدعي الدالة =minverse

SUM		X		✓		fx		=MINVERSE(A2:D5)		
	A	B	C	D	E	F	G	H	I	J
1	x1	x2	x3	x4	b					
2	1	-1	2	-1	-8	=MINVERSE(A2:D5)				
3	2	-2	3	-3	-20					
4	1	1	1	0	-2					
5	1	-1	4	3	4					
6										

وبعد تحديد نطاق المصفوفة الأصلية A2:D5 المراد حساب نظيرها نضغط معا

Ctrl+shift+enter

فنحصل على المصفوفة العكسية كما يلي

G2		fx {=MINVERSE(A2:D5)}								
	A	B	C	D	E	F	G	H	I	J
1	x1	x2	x3	x4	b					
2	1	-1	2	-1	-8		-7.5	3.5	0.5	1
3	2	-2	3	-3	-20		3	-1.5	0.5	-0.5
4	1	1	1	0	-2		4.5	-2	0	-0.5
5	1	-1	4	3	4		-2.5	1	0	0.5

نضرب المصفوفة العكسية بالطرف الأيمن للمنظومة فنحصل على الحل (ظل منطقة الحل وهي النطاق (L2:L5)

SUM													
	A	B	C	D	E	F	G	H	I	J	K	L	M
1	x1	x2	x3	x4	b							x=	
2	1	-1	2	-1	-8		-7.5	3.5	0.5	1	=MMULT(G2:J5,E2:E5)		
3	2	-2	3	-3	-20		3	-1.5	0.5	-0.5			
4	1	1	1	0	-2		4.5	-2	0	-0.5			
5	1	-1	4	3	4		-2.5	1	0	0.5			

وعמוד الحل يكون بعد أن نكتب ctrl+shift+enter كما يلي

L5													
	A	B	C	D	E	F	G	H	I	J	K	L	M
1	x1	x2	x3	x4	b							x=	
2	1	-1	2	-1	-8		-7.5	3.5	0.5	1	x1=	-7	
3	2	-2	3	-3	-20		3	-1.5	0.5	-0.5	x2=	3	
4	1	1	1	0	-2		4.5	-2	0	-0.5	x3=	2	
5	1	-1	4	3	4		-2.5	1	0	0.5	x4=	2	

طريقة حذف جاوس

أما إذا أردنا استخدام طريقة حذف جاوس

K15													
	A	B	C	D	E	F	G	H	I	J	K	L	M
1	-1	x1	x2	x3	x4	b		-2	x1	x2	x3	x4	b
2	E1	1	-1	2	-1	-8		E1	1	-1	2	-1	-8
3	E2	2	-2	3	-3	-20		E2-2*E1	0	0	-1	-1	-4
4	E3	1	1	1	0	-2		E3-E1	0	2	-1	1	6
5	E4	1	-1	4	3	4		E4-E1	0	0	2	4	12
6													
7	-3	x1	x2	x3	x4	b		-4	x1	x2	x3	x4	b
8	E1	1	-1	2	-1	-8		E1	1	-1	2	-1	-8
9	E2	0	2	-1	1	6		E2	0	2	-1	1	6
10	E3	0	0	-1	-1	-4		E3	0	0	-1	-1	-4
11	E4+2*E3	0	0	2	4	12		E4	0	0	0	2	4
12													
13													
14													
15	x4=	2		x3=	2		x3=	3		x1=	-7		

بالتعويض العكسي

لاحظ لطرح صف من صف نكتب كما يلي:-

9	E2	0	2	-1	1	6	E2	0	2	-1	1	6
10	E3	0	0	-1	-1	-4	E3	0	0	-1	-1	-4
11	E4+2*E3	0	0	2	4	12	=B11:F11+2*B10:F10	0	2	4		
12												

ثم نستعمل المفاتيح ctrl+shift+enter معا

ومعادلات التعويض العكسي كما يلي:

13						بالتعويض العكسي			
14									
15	x4=	=M11/L11	x3=	=(M10+B15)	x3=	=(M9-B15+E15)/2	x1=	=M8+B15-2*E15+H15	

الطرق الغير مباشرة لحل منظومة معادلات خطية

طريقة جاكوبي

$$4x_1 + x_2 - 4x_3 = -5$$

$$x_1 + 2x_2 + x_3 = 1$$

$$-x_2 + 4x_3 = 9$$

نحول المعادلات إلى الشكل الآتي

$$x_1 = \frac{-5 - x_2 + 4x_3}{4}$$

$$x_2 = \frac{1 - x_1 - x_3}{2}$$

$$x_3 = \frac{9 + x_2}{4}$$

	A	B	C
1		Jacobi	
2	x1=	x2=	x3=
3	1	1	1
4	=(-5-B3+4*C3)/4	=(1-A3-C3)/2	=(9+B3)/4

وبعد تنفيذ الشكل السابق من المعادلات ونسخها للأسفل نحصل على

	A	B	C	D
1		Jacobi		
2	x1=	x2=	x3=	
3	1	1	1	
4	-0.5	-0.5	2.5	
5	1.375	-0.5	2.125	
6	1	-1.25	2.125	
7	1.1875	-1.0625	1.9375	
8	0.953125	-1.0625	1.984375	
9	1	-0.96875	1.984375	
10	0.976563	-0.99219	2.007813	
11	1.005859	-0.99219	2.001953	
12	1	-1.00391	2.001953	

طريق جاوس سايدل

نقوم بتعديل المعادلات في الورقة لتصبح كما يلي

	A	B	C	
1		Gauss Seidel		
2	x1=	x2=	x3=	
3	1	1	1	
4	$=(-5-B3+4*C3)/4$	$=(1-A4-C3)/2$	$=(9+B4)/4$	
5	$=(-5-B4+4*C4)/4$	$=(1-A5-C4)/2$	$=(9+B5)/4$	

وعند التطبيق نحصل على حل أسرع من طريقة جاكوبي أي بعدد أقل من التكرار

	A	B	C	D
1		Gauss Seidel		
2	x1=	x2=	x3=	
3	1	1	1	
4	-0.5	0.25	2.3125	
5	1	-1.15625	1.960938	
6	1	-0.98047	2.004883	
7	1	-1.00244	1.99939	
8	1	-0.99969	2.000076	

طريقة SOR

تعتمد هذه الطريقة على جاوس سايدل ولو أخذنا $w=1.2$

	A	B	C	D	E	F	G
1		Gauss Seidel			SOR	w=	1.5
2	x1=	x2=	x3=		x1	x2	x3
3	1	1	1				
4	$=(-5-B3+4*C3)/4$	$=(1-A4-C3)/2$	$=(9+B4)/4$		$=(1-G$1)*A3+G$1*A4$	$=(1-G$1)*B3+G$1*B4$	$=(1-G$1)*C3+G$1*C4$
5	$=(-5-B4+4*C4)/4$	$=(1-A5-C4)/2$	$=(9+B5)/4$		$=(1-G$1)*A4+G$1*A5$	$=(1-G$1)*B4+G$1*B5$	$=(1-G$1)*C4+G$1*C5$
6	$=(-5-B5+4*C5)/4$	$=(1-A6-C5)/2$	$=(9+B6)/4$		$=(1-G$1)*A5+G$1*A6$	$=(1-G$1)*B5+G$1*B6$	$=(1-G$1)*C5+G$1*C6$
7	$=(-5-B6+4*C6)/4$	$=(1-A7-C6)/2$	$=(9+B7)/4$		$=(1-G$1)*A6+G$1*A7$	$=(1-G$1)*B6+G$1*B7$	$=(1-G$1)*C6+G$1*C7$
8	$=(-5-B7+4*C7)/4$	$=(1-A8-C7)/2$	$=(9+B8)/4$		$=(1-G$1)*A7+G$1*A8$	$=(1-G$1)*B7+G$1*B8$	$=(1-G$1)*C7+G$1*C8$

والقيم الرقمية هي

	A	B	C	D	E	F	G
1		Gauss Seidel			SOR	w=	1.2
2	x1=	x2=	x3=		x1	x2	x3
3	1	1	1				
4	-0.5	0.25	2.313		-0.8	0.1	2.575
5	1	-1.156	1.961		1.3	-1.4375	1.890625
6	1	-0.98	2.005		1	-0.9453125	2.01367188
7	1	-1.002	1.999		1	-1.0068359	1.99829102
8	1	-1	2		1	-0.9991455	2.00021362

طريقة نيوتن لمنظومة معادلات غير خطية من

$$x + 2y = 4$$

$$2x^2 + y^2 = 7$$

الحل:

نكتب المعادلات بالشكل الآتي

$$f_1(x, y) = x + 2y - 4$$

$$f_2(x, y) = 2x^2 + y^2 - 7$$

نجد الجاكوبي:

$$J(x, y) = \begin{bmatrix} \frac{\partial f_1}{\partial x} & \frac{\partial f_1}{\partial y} \\ \frac{\partial f_2}{\partial x} & \frac{\partial f_2}{\partial y} \end{bmatrix} = \begin{bmatrix} 1 & 2 \\ 4x & 2y \end{bmatrix}$$

$$|J(x, y)| = \begin{vmatrix} \frac{\partial f_1}{\partial x} & \frac{\partial f_1}{\partial y} \\ \frac{\partial f_2}{\partial x} & \frac{\partial f_2}{\partial y} \end{vmatrix} = \begin{vmatrix} 1 & 2 \\ 4x & 2y \end{vmatrix} = 2y - 8x$$

$$J^{-1} = \frac{1}{(2y - 8x)} \begin{bmatrix} 2y & -2 \\ -4x & 1 \end{bmatrix}$$

نكتب صيغة نيوتن

$$\begin{bmatrix} x_{n+1} \\ y_{n+1} \end{bmatrix} = \begin{bmatrix} x_n \\ y_n \end{bmatrix} - J^{-1} \begin{bmatrix} f_1(x_n, y_n) \\ f_2(x_n, y_n) \end{bmatrix}$$

$$= \begin{bmatrix} x_n \\ y_n \end{bmatrix} - \frac{1}{(2y_n - 8x_n)} \begin{bmatrix} 2y_n & -2 \\ -4x_n & 1 \end{bmatrix} \begin{bmatrix} x_n + 2y_n - 4 \\ 2x_n^2 + y_n^2 - 7 \end{bmatrix}$$

ولحل المنظومة باستخدام إكسل علينا كتابة الصيغة العامة

$$\vec{X}_{n+1} = \vec{X}_n + \Delta \vec{X}, \quad \Delta \vec{X} = -J^{-1} \times F(\vec{X}_n)$$

نفرض القيم الأولية هي

$$\vec{X}_0 = [1.5, \quad 1.5]^T$$

ثم نحسب الدالة F عند النقطة السابقة

$$F(\vec{X}_n) = \begin{bmatrix} f_1(x_0, y_0) \\ f_2(x_0, y_0) \end{bmatrix} = \begin{bmatrix} x_n + 2y_n - 4 \\ 2x_n^2 + y_n^2 - 7 \end{bmatrix} = \begin{bmatrix} 0.5 \\ -0.25 \end{bmatrix}$$

	A	B	E
1	step	f1=x+2y-4	f2=2x^2+y^2-7
2		Xi	F(Xi)
3		1.5	=B3+2*B4-4
4	1	1.5	=2*B3^2+B4^2-7

نحسب مصفوف الجاكوبي

$$J(x, y) = \begin{bmatrix} 1 & 2 \\ 4x & 2y \end{bmatrix}$$

	A	B	C	D	E	F	G	H
1	step	f1=x2+y-4		f2=2x^2+y^2-7		1	2	هذه قيم ثابتة في مصفوفة جاكوبي
2		Xi	J=		F(Xi)			
3		1.5	=F\$1	=G\$1	=B3^2+B4-4			
4	1	1.5	=4*B3	=2*B4	=2*B3^2+B4^2-7			

والقيم العددية لهذه الخطوة (لاحظ أن قيم الصف الأول قيم ثابتة وهي 1 و 2 لهذا تم الإشارة لها بمرجعية مطلقة أي \$F\$1 ، و \$G\$1)

	A	B	C	D	E
1	step	f1=x+2y-4		f2=2x^2+y^2-7	f2=2x^2+y^2-7
2		Xi		J=	F(Xi)
3		1.5	1	2	0.5
4	1	1.5	6	3	-0.25

نحسب قيم

$$\Delta \vec{X} = -J^{-1} \times F(\vec{X}_n)$$

كما ننقل القيم الجديدة ل X في العمود الأول كما يظهر في الورقة

	A	B	C	D	E	F	
1	step	f1=x+2y-4		f2=2x^2+y^2-7	f2=2x^2+y^2-7	1	2
2		Xi		J=	F(Xi)		
3		1.5	=F\$1	=G\$1	=B3+2*B4-4	=MMULT(MINVERSE(C3:D4),E3:E4)	
4	1	1.5	=4*B3	=2*B4	=2*B3^2+B4^2-7	=MMULT(MINVERSE(C3:D4),E3:E4)	
5		=B3:B4+F3:F4					
6	2	=B3:B4+F3:F4					

والنتائج لهذه المعادلات هي

	A	B	C	D	E	F	G
1	step	f1=x+2y-4		f2=2x^2+y^2-7	f2=2x^2+y^2-7	1	2
2		Xi		J=	F(Xi)		
3		1.5	1	2	0.5	0.222222222	
4	1	1.5	6	3	-0.25	-0.361111111	
5		1.722222					
6	2	1.138889					

نسحب لبقية القيم

	A	B	C	D	E	F	G
1	step	f1=x+2y-4		f2=2x^2+y^2-7		1	2
2		Xi		J=	F(Xi)		
3		1.5	1	2	0.5	-0.222222222	
4	1	1.5	6	3	-0.25	0.361111111	
5		1.722222222	1	2	0	0.039855072	
6	2	1.138888889	6.888889	2.277777778	0.2291667	-0.019927536	
7		1.68236715	1	2	0	0.000641569	
8	3	1.158816425	6.729469	2.31763285	0.003574	-0.000320785	
9		1.68172558	1	2	0	1.66337E-07	
10	4	1.15913721	6.726902	2.31827442	9.261E-07	-8.31685E-08	
11		1.681725414					
12	5	1.159137293					

مثال: حل النظام الآتي بطريقة نيوتن

$$\begin{aligned} 4x^2 - y^2 &= 0 \\ 4xy^2 - x &= 1 \end{aligned}$$

الحل:

نكتب المعادلات بالشكل الآتي

$$\begin{aligned} f_1(x, y) &= 4x^2 - y^2 \\ f_2(x, y) &= 4xy^2 - x - 1 \end{aligned}$$

نجد الجاكوبي:

$$J(x, y) = \begin{bmatrix} \frac{\partial f_1}{\partial x} & \frac{\partial f_1}{\partial y} \\ \frac{\partial f_2}{\partial x} & \frac{\partial f_2}{\partial y} \end{bmatrix} = \begin{bmatrix} 8x & -2y \\ 4y^2 - 1 & 8xy \end{bmatrix}$$

ولحل المنظومة باستخدام إكسل علينا كتابة الصيغة العامة

$$\overrightarrow{X_{n+1}} = \overrightarrow{X_n} + \overrightarrow{\Delta X}, \quad \overrightarrow{\Delta X} = -J^{-1} \times F(\overrightarrow{X_n})$$

نفرض القيم الأولية هي

$$\overrightarrow{X_0} = [0, \quad 1]^T$$

ندخل المعادلات في ورقة العمل

	A	B	C	D	E	F
1	step	f1=4x^2-y^2		f2=4xy^2-xy^2-1		
2		Xi		J=	F(Xi)	
3		0	=8*B3	=-2*B4	=4*B3^2-B4^2	=-MMULT(MINVERSE(C3:D4),E3:E4)
4	1	1	=4*B4^2-1	=8*B3*B4	=4*B3*B4^2-B3-1	=-MMULT(MINVERSE(C3:D4),E3:E4)
5		=B3:B4+F3:F4				
6	2	=B3:B4+F3:F4				

القيم العددية هي

	A	B	C	D	E	F
1	step	f1=4x^2-y^2		f2=4xy^2-xy^2-1		
2		Xi		J=	F(Xi)	
3		0	0	-2	-1	0.33333333
4	1	1	3	0	-1	-0.5
5		0.33333333				
6	2	0.5				

نسحب لحساب باقي القيم فنجد أن قيم الحل النهائي هي

$$\overrightarrow{X_n} = [0.44909, \quad 0.898192]^T$$

F3		fx {=MMULT(MINVERSE(C3:D4),E3:E4)}					
	A	B	C	D	E	F	G
1	step	f1=4x^2-y^2		f2=4xy^2-xy^2-1			
2		Xi		J=	F(Xi)	inver(J)*F(X)	
3		0	0	-2	-1	0.333333333	
4	1	1	3	0	-1	-0.5	
5		0.333333333	2.666667	-1	0.19444	0.208333333	
6	2	0.5	0	1.333333333	-1	0.75	
7		0.54166667	4.333333	-2.5	-0.38889	-0.068390259	
8	3	1.25	5.25	5.416666667	1.84375	-0.274098672	
9		0.47327641	3.786211	-1.951802657	-0.05642	-0.022338335	
10	4	0.97590133	2.809534	3.694968596	0.32969	-0.072240308	
11		0.45093807	3.607505	-1.80732204	-0.00322	-0.001846366	
12	5	0.90366102	2.266413	3.259961269	0.02201	-0.00546855	
13		0.44909171	3.592734	-1.796384939	-1.6E-05	-1.12304E-05	
14		0.89819247	2.226999	3.226966312	0.00013	-3.15169E-05	

استعمال Solver لحل المنظومة الغير خطية من المعادلات
مثال: أوجد حل المنظومة الآتية

$$u(x, y) = x^2 + xy - 10 = 0$$

$$v(x, y) = y + 3xy^2 - 57 = 0$$

$$x_0 = 1, \quad y_0 = 1$$


ندخل الحل الأولي والمعادلات أعلاه في ورقة اكسل، ثم ندخل المقدار من أجل وضع المقدار الآتي صفراً

$$Z = u(x, y)^2 + v(x, y)^2$$


B5		fx =B3^2+B4^2	
	A	B	C
1	x=	1	
2	y=	1	
3	f1	=B1^2+B1*B2-10	
4	f2	=B2+3*B1*B2^2-57	
5	Z=	=B3^2+B4^2	

نستدعي Solver من تبويب Data

نشاط: عندما لا يكون Solver موجودا في تبويب Data نذهب إلى قائمة File ونختار Options ومن ثم
Go بعدها add-in



ثم نختار Solver add-in من القائمة الجديدة



فنجد أن Solver قد اندرج في تبويب Data
نستدعي Solver ونختار الهدف وهو جعل قيمة $Z=0$ وذلك بتغيير القيم الأولية للحل

Solver Parameters

Set Objective:

To: ☐ Max ☐ Min ☒ Value Of:

By Changing Variable Cells:

Subject to the Constraints:

☒ Make Unconstrained Variables Non-Negative

Select a Solving Method:

Solving Method
Select the GRG Nonlinear engine for Solver Problems that are smooth nonlinear. Select the LP Simplex engine for linear Solver Problems, and select the Evolutionary engine for Solver problems that are non-smooth.

فيكون الحل

	B5	f_x	=B3^2+B4^2		
	A	B	C	D	E
1	x=	2.000541			
2	y=	3.00041			
3	f1	0.004607			
4	f2=	0.029772			
5	Z=	0.000908			

والجواب

$$x_0 = 2.000541, \quad y_0 = 3.00041$$

طريقة النقطة الثابتة

مثال: حل طريقة النقطة الثابتة لمنظومة معادلات غير خطية الآتية

$$x + 2y = 4$$

$$2x^2 + y^2 = 7$$

$$D = \{(x, y) | 1 \leq x \leq 2, 1 \leq y \leq 2\}$$

نحول المعادلات إلى الشكل الآتي

$$x = g_1(x, y) = \sqrt{\frac{7-y^2}{2}}, \quad y = g_2(x, y) = \frac{4-x}{2}$$

ندخل البيانات على ورقة إكسل

	A	B	C	D
1	x	y	g1	g2
2	1.5	1.5	=SQRT((7-B2^2)/2)	=(4-A2)/2
3	=C2	=D2		

ونسحب للباقي

	A	B	C	D
1	x	y	g1	g2
2	1.5	1.5	1.5411035	1.25
3	1.541104	1.25	1.6488632	1.229448
4	1.648863	1.2294482	1.6565713	1.175568
5	1.656571	1.1755684	1.676013	1.171714
6	1.676013	1.1717143	1.6773618	1.161994
7	1.677362	1.1619935	1.6807396	1.161319
8	1.68074	1.1613191	1.6809726	1.15963
9	1.680973	1.1596302		

المجموع الأدنى للمربعات
التقريب الخطي للبيانات
ليكن لدينا البيانات الآتية

x	6	8	10	12	14	16	18
y	3.8	3.7	4	3.9	4.3	4.2	4.2

	A	B	C	D
1	x	y	xy	x^2
2	6	3.8	=A2*B2	=A2^2
3	8	3.7	=A3*B3	=A3^2
4	10	4	=A4*B4	=A4^2
5	12	3.9	=A5*B5	=A5^2
6	14	4.3	=A6*B6	=A6^2
7	16	4.2	=A7*B7	=A7^2
8	18	4.2	=A8*B8	=A8^2
9	=SUM(A2:A8)	=SUM(B2:B8)	=SUM(C2:C8)	=SUM(D2:D8)

والقيمة العددية للبيانات

	A	B	C	D	E	F	G	H	I	J
1	x	y	xy	x^2			B	M		T
2	6	3.8	22.8	36			7	84		28.1
3	8	3.7	29.6	64			84	1120		342.2
4	10	4	40	100						
5	12	3.9	46.8	144			1.428571	-0.10714		3.478571
6	14	4.3	60.2	196			-0.10714	0.008929		0.044643
7	16	4.2	67.2	256						
8	18	4.2	75.6	324			B=3.478571			
9	84	28.1	342.2	1120			M=0.044643			

التكاملات العددية
أولاً: قاعدة منتصف الفترة
نفرض التكامل الآتي

$$\int_a^b f(x) dx$$

نقرب الاقتران على أنه مستطيل قاعدته هي حدود التكامل $b-a$ والارتفاع هو قيمة الاقتران عن منتصف الفترة فإن التكامل العددي يصبح

$$\int_a^b f(x) dx = (b-a) f\left(\frac{a+b}{2}\right)$$

إذا تم تجزئة الفترة إلى m من الفترات فطول قاعدة كل فترة واحداثيات النقاط تصبح

$$h = \frac{b-a}{m}, \quad x_j = a + j h, \quad j = 0, 1, 2, \dots, m$$

$$\int_a^b f(x) dx = h \sum_{j=0}^{m-1} f\left(\frac{x_j + x_{j+1}}{2}\right)$$

مثال: أوجد التكامل الآتي على ست فترات

$$\int_0^3 x^2 dx$$

ندرج المعادلات الآتية وفق قوانين التجزئة

	A	B	C	D	E	F	G	H	I
1	a=	0	b=	3	m=	6	h=	=(D1-B1)/F1	ف الفترة
2									
3	j=	x	Xc=	f(x)=					
4	0	=B\$1+A4*\$H\$1		=(B4+B5)/2^2	Integ=	=H1*D11			
5	1	=B\$1+A5*\$H\$1	=(B5+B4)/2	=(B5+B6)/2^2					
6	2	=B\$1+A6*\$H\$1	=(B6+B5)/2	=(B6+B7)/2^2					
7	3	=B\$1+A7*\$H\$1	=(B7+B6)/2	=(B7+B8)/2^2					
8	4	=B\$1+A8*\$H\$1	=(B8+B7)/2	=(B8+B9)/2^2					
9	5	=B\$1+A9*\$H\$1	=(B9+B8)/2	=(B9+B10)/2^2					
10	6	=B\$1+A10*\$H\$1	=(B10+B9)/2	=(B10+B11)/2^2					
11				=SUM(D4:D10)					

وبعد تطبيق المعادلات نجد أن التكامل هو

$$\int_0^3 x^2 dx = 0.5 \left(f\left(\frac{0+0.5}{2}\right) + f(0.75) + f(1.25) + f(1.75) + f(2.25) + f(2.75) \right) = 10.0625$$

	F4			f_x	=H1*D11				
	A	B	C	D	E	F	G	H	I
1	a=	0	b=	3	m=	6	h=	0.5	منتصف الفترة
2									
3	j=	x	Xc=	f(x)=					
4	0	0		0.0625	Integ=	10.0625			
5	1	0.5	0.25	0.5625					
6	2	1	0.75	1.5625					
7	3	1.5	1.25	3.0625					
8	4	2	1.75	5.0625					
9	5	2.5	2.25	7.5625					
10	6	3	2.75	2.25					
11				20.125					

ثانياً: قاعدة شبه المنحرف
نفرض التكامل الآتي

$$\int_a^b f(x) dx$$

نقرب الأقتران على أنه شبه منحرف قاعدته هي قيمة الاقتران عند حدود التكامل والارتفاع هو حدود التكامل

$$\int_a^b f(x) dx = \left(\frac{(b-a)}{2} \right) [f(a) + f(b)] = \frac{h}{2} [f(a) + f(b)]$$

$$error = E = -\frac{h^3}{12} f''(c), \quad c \in [a, b]$$

إذا تم تجزئة الفترة إلى m من الفترات فطول قاعدة كل فترة واحداثيات النقاط تصبح

$$h = \frac{b-a}{m}, \quad x_j = a + j h, \quad j = 1, 2, \dots, m-1$$

$$\int_a^b f(x) dx = \left(\frac{h}{2} \right) [f(a) + f(b) + 2 \sum_{j=1}^{m-1} f(x_j)]$$

مثال: أوجد التكامل الآتي على ست فترات

$$\int_0^3 x^2 dx$$

ندرج المعادلات الآتية وفق قوانين التجزئة

	A	B	C	D	E	F	G
1	a=	0	b=	3	m=	6	
2	f(a)=	=B1^2	f(b)=	=D1^2	h=	=(D1-B1)/F1	تبيه المنحرف
3	j=	x	f(x)=				
4	1	=\$B\$1+A4*\$F\$2	=B4^2		Integ=	=F2/2*(B2+D2+2*C9)	
5	2	=\$B\$1+A5*\$F\$2	=B5^2				
6	3	=\$B\$1+A6*\$F\$2	=B6^2				
7	4	=\$B\$1+A7*\$F\$2	=B7^2				
8	5	=\$B\$1+A8*\$F\$2	=B8^2				
9			=SUM(C4:C8)				

والحسابات هي

F4		fx		=F2/2*(B2+D2+2*C9)			
	A	B	C	D	E	F	G
1	a=	0	b=	3	m=	6	
2	f(a)=	0	f(b)=	9	h=	0.5	تبيه المنحرف
3	j=	x	f(x)=				
4	1	0.5	0.25		Integ=	9.125	
5	2	1	1				
6	3	1.5	2.25				
7	4	2	4				
8	5	2.5	6.25				
9			13.75				

ثانياً: قاعدة سيمبسون
نفرض التكامل الآتي

$$\int_a^b f(x)dx$$

نقرب الأقران على بحيث يتم تجزئة كل فترة إلى فترتين

$$h = \frac{b-a}{2}, \quad x_0 = a, \quad x_2 = b, \quad x_1 = \frac{a+b}{2}$$

$$\int_a^b f(x)dx = \frac{h}{3} (f(x_0) + 4f(x_1) + f(x_2))$$

$$error = E = -\frac{h^2}{90} f^{(4)}(c), \quad c \in [a, b]$$

إذا تم تجزئة الفترة إلى m من الفترات فطول قاعدة كل فترة واحداثيات النقاط تصبح

$$h = \frac{b-a}{2m}, \quad x_j = a + jh, \quad j = 1, 2, \dots, m-1$$

$$\int_a^b f(x)dx = \left(\frac{h}{3}\right) \left(f(a) + f(b) + 2 \sum_{j=1}^{m-1} f(x_{2j}) + 4 \sum_{j=1}^m f(x_{2j-1}) \right)$$

مثال: أوجد التكامل الآتي على عشرة فترات الفرعية

$$\int_1^3 \frac{x}{1+x^2} dx$$

$$2m = 10$$

ندرج المعادلات الآتية وفق قوانين التجزئة

C14		fx =H1/3*(B1/((1+B1^2)+D1/((1+D1^2)+2*D13+4*E13)						
A	B	C	D	E	F	G	H	
1	a=	1	b=	3	2m=	10	h=	=(D1-B1)/F1
2								
3	i=	x	f(x)=	even	odd			
4	1	=B\$1+A4*\$H\$1	=B4/(1+B4^2)	=IF(MOD(A4,2)=0,C4,0)	=IF(MOD(A4,2)=1,C4,0)			
5	2	=B\$1+A5*\$H\$1	=B5/(1+B5^2)	=IF(MOD(A5,2)=0,C5,0)	=IF(MOD(A5,2)=1,C5,0)			
6	3	=B\$1+A6*\$H\$1	=B6/(1+B6^2)	=IF(MOD(A6,2)=0,C6,0)	=IF(MOD(A6,2)=1,C6,0)			
7	4	=B\$1+A7*\$H\$1	=B7/(1+B7^2)	=IF(MOD(A7,2)=0,C7,0)	=IF(MOD(A7,2)=1,C7,0)			
8	5	=B\$1+A8*\$H\$1	=B8/(1+B8^2)	=IF(MOD(A8,2)=0,C8,0)	=IF(MOD(A8,2)=1,C8,0)			
9	6	=B\$1+A9*\$H\$1	=B9/(1+B9^2)	=IF(MOD(A9,2)=0,C9,0)	=IF(MOD(A9,2)=1,C9,0)			
10	7	=B\$1+A10*\$H\$1	=B10/(1+B10^2)	=IF(MOD(A10,2)=0,C10,0)	=IF(MOD(A10,2)=1,C10,0)			
11	8	=B\$1+A11*\$H\$1	=B11/(1+B11^2)	=IF(MOD(A11,2)=0,C11,0)	=IF(MOD(A11,2)=1,C11,0)			
12	9	=B\$1+A12*\$H\$1	=B12/(1+B12^2)	=IF(MOD(A12,2)=0,C12,0)	=IF(MOD(A12,2)=1,C12,0)			
13			=SUM(D4:D12)	=SUM(E4:E12)				
14	Integ=	=H1/3*(B1/(1+B1^2)						

ومنه

C14		fx		=H1/3*(B1/(1+B1^2)+D1/(1+D1^2)+2				
A	B	C	D	E	F	G	H	
1	a=	1	b=	3	2m=	10	h=	0.2
2								
3	i=	x	f(x)=	even	odd			
4	1	1.2	0.49180328	0	0.49180328			
5	2	1.4	0.47297297	0.47297297	0			
6	3	1.6	0.4494382	0	0.4494382			
7	4	1.8	0.4245283	0.4245283	0			
8	5	2	0.4	0	0.4			
9	6	2.2	0.37671233	0.37671233	0			
10	7	2.4	0.35502959	0	0.35502959			
11	8	2.6	0.33505155	0.33505155	0			
12	9	2.8	0.31674208	0	0.31674208			
13				1.60926515	2.01301315			
14	Integ=	0.80470553						